NAG Fortran Library Routine Document

F07NWF (CSYTRI/ZSYTRI)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F07NWF (CSYTRI/ZSYTRI) computes the inverse of a complex symmetric matrix A, where A has been factorized by F07NRF (CSYTRF/ZSYTRF).

2 Specification

```
SUBROUTINE F07NWF(UPLO, N, A, LDA, IPIV, WORK, INFO)
ENTRY      csytri  (UPLO, N, A, LDA, IPIV, WORK, INFO)
INTEGER            N, LDA, IPIV(*), INFO
complex            A(LDA,*), WORK(*)
CHARACTER*1        UPLO
```

The ENTRY statement enables the routine to be called by its LAPACK name.

3 Description

To compute the inverse of a complex symmetric matrix A, this routine must be preceded by a call to F07NRF (CSYTRF/ZSYTRF), which computes the Bunch–Kaufman factorization of A.

If UPLO = 'U', $A = PUDU^TP^T$ and A^{-1} is computed by solving $U^TP^TXPU = D^{-1}$ for X.

If UPLO = 'L', $A = PLDL^TP^T$ and A^{-1} is computed by solving $L^TP^TXPL = D^{-1}$ for X.

4 References

Du Croz J J and Higham N J (1992) Stability of methods for matrix inversion *IMA J. Numer. Anal.* **12** 1–19

5 Parameters

1: UPLO – CHARACTER*1 *Input*

On entry: indicates how A has been factorized as follows:

if UPLO = 'U', $A = PUDU^TP^T$, where U is upper triangular;

if UPLO = 'L', $A = PLDL^TP^T$, where L is lower triangular.

Constraint: UPLO = 'U' or 'L'.

2: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: N ≥ 0.

3: A(LDA,*) – ***complex*** array *Input/Output*

Note: the second dimension of the array A must be at least $\max(1, \mathrm{N})$.

On entry: details of the factorization of A, as returned by F07NRF (CSYTRF/ZSYTRF).

On exit: the factorization is overwritten by the n by n symmetric matrix A^{-1}. If UPLO='U', the upper triangle of A^{-1} is stored in the upper triangular part of the array; if UPLO = 'L', the lower triangle of A^{-1} is stored in the lower triangular part of the array.

4: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F07NWF (CSYTRI/ZSYTRI) is called.

Constraint: LDA $\geq \max(1, \mathrm{N})$.

5: IPIV(∗) – INTEGER array *Input*

Note: the dimension of the array IPIV must be at least $\max(1, \mathrm{N})$.

On entry: details of the interchanges and the block structure of D, as returned by F07NRF (CSYTRF/ZSYTRF).

6: WORK(∗) – ***complex*** array *Workspace*

Note: the dimension of the array WORK must be at least $\max(1, 2 * \mathrm{N})$.

7: INFO – INTEGER *Output*

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO $= -i$, the ith parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

If INFO $= i$, d_{ii} is exactly zero; D is singular and the inverse of A cannot be computed.

7 Accuracy

The computed inverse X satisfies a bound of the form

$|DU^T P^T XPU - I| \leq c(n)\epsilon(|D|\,|U^T|P^T|X|P|U| + |D|\,|D^{-1}|)$, if UPLO = 'U', or

$|DL^T P^T XPL - I| \leq c(n)\epsilon(|D|\,|L^T|P^T|X|P|L| + |D|\,|D^{-1}|)$, if UPLO = 'L',

where $c(n)$ is a modest linear function of n, and ϵ is the ***machine precision***.

8 Further Comments

The total number of real floating-point operations is approximately $\frac{8}{3}n^3$.

The real analogue of this routine is F07MJF (SSYTRI/DSYTRI).

9 Example

To compute the inverse of the matrix A, where

$$A = \begin{pmatrix} -0.39 - 0.71i & 5.14 - 0.64i & -7.86 - 2.96i & 3.80 + 0.92i \\ 5.14 - 0.64i & 8.86 + 1.81i & -3.52 + 0.58i & 5.32 - 1.59i \\ -7.86 - 2.96i & -3.52 + 0.58i & -2.83 - 0.03i & -1.54 - 2.86i \\ 3.80 + 0.92i & 5.32 - 1.59i & -1.54 - 2.86i & -0.56 + 0.12i \end{pmatrix}.$$

Here A is symmetric and must first be factorized by F07NRF (CSYTRF/ZSYTRF).

9.1 Program Text

Note: the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     F07NWF Example Program Text
*     Mark 15 Release. NAG Copyright 1991.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX, LDA, LWORK
      PARAMETER        (NMAX=8,LDA=NMAX,LWORK=64*NMAX)
*     .. Local Scalars ..
      INTEGER          I, IFAIL, INFO, J, N
      CHARACTER        UPLO
*     .. Local Arrays ..
      complex          A(LDA,NMAX), WORK(LWORK)
      INTEGER          IPIV(NMAX)
      CHARACTER        CLABS(1), RLABS(1)
*     .. External Subroutines ..
      EXTERNAL         csytrf, csytri, X04DBF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F07NWF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      IF (N.LE.NMAX) THEN
*
*        Read A from data file
*
         READ (NIN,*) UPLO
         IF (UPLO.EQ.'U') THEN
            READ (NIN,*) ((A(I,J),J=I,N),I=1,N)
         ELSE IF (UPLO.EQ.'L') THEN
            READ (NIN,*) ((A(I,J),J=1,I),I=1,N)
         END IF
*
*        Factorize A
*
         CALL csytrf(UPLO,N,A,LDA,IPIV,WORK,LWORK,INFO)
*
         WRITE (NOUT,*)
         IF (INFO.EQ.0) THEN
*
*           Compute inverse of A
*
            CALL csytri(UPLO,N,A,LDA,IPIV,WORK,INFO)
*
*           Print inverse
*
            IFAIL = 0
            CALL X04DBF(UPLO,'Nonunit',N,N,A,LDA,'Bracketed','F7.4',
     +                  'Inverse','Integer',RLABS,'Integer',CLABS,80,0,
     +                  IFAIL)
         ELSE
            WRITE (NOUT,*) 'The factor D is singular'
         END IF
       END IF
```

```
      STOP
*
      END
```

9.2 Program Data

```
F07NWF Example Program Data
  4                                                          :Value of N
  'L'                                                        :Value of UPLO
 (-0.39,-0.71)
 ( 5.14,-0.64) ( 8.86, 1.81)
 (-7.86,-2.96) (-3.52, 0.58) (-2.83,-0.03)
 ( 3.80, 0.92) ( 5.32,-1.59) (-1.54,-2.86) (-0.56, 0.12)  :End of matrix A
```

9.3 Program Results

```
 F07NWF Example Program Results

 Inverse
                     1                 2                 3                 4
 1  (-0.1562,-0.1014)
 2  ( 0.0400, 0.1527) ( 0.0946,-0.1475)
 3  ( 0.0550, 0.0845) (-0.0326,-0.1370) (-0.1320,-0.0102)
 4  ( 0.2162,-0.0742) (-0.0995,-0.0461) (-0.1793, 0.1183) (-0.2269, 0.2383)
```